
'// Rutina para envío de comandos a la impresora fiscal.
'//
'//     nrohand   = Manejador del puerto serie ( handler devuelto por la función de apertura del puerto serie )
'//     strcmd     = Comando a enviar ( armado por el software )
'//=========================================================================================
Public Sub EnviarComando(nrohand As Long, strcmd As String)
Static Atomic As Boolean                                                                               '// Control operaciones atómicas
Dim cod As Long                                                                                            '// Código de retorno función DLL
Dim aux As Integer                                                                                        '//  Variable auxiliar

    '// Si todavía no se ha recibido y procesado la respuesta al comando anterior,
    '// rechazar el envío del comando actual y abandonar la subrutina
    '// --------------------------------------------------------------------------------------------
    If (Atomic) Then
        MsgBox "ERROR:" & vbCrLf & "No puede ejecutar más de un comando por vez...", vbCritical, "Error..."
        '// Ver que hacer de acuerdo a la lógica del software
        Exit Sub
    End If

    '// Se hace atómica la operación de procesado de la respuesta,
    '// y se habilitan mensajes al usuario
    '// -------------------------------------------------------------------------
    Atomic = True
    FrmMain.Timer2.Enabled = True

    Do
        '// Se envía comando a impresora fiscal
        '// ---------------------------------------------
        cod = MandaPaqueteFiscal(nrohand, strcmd)
                                                              
        If ((cod < 0) And (cod <> ERR_ATOMIC)) Then
            '// La función de la DLL retornó Error
            '// -------------------------------------------
            MsgBox "ERROR:  " & Str(cod) & "Al Enviar Comando..!!" & vbCrLf & vbCrLf & _
                         "Comando = " & CambiarSep(strcmd), vbCritical, "Error..."
            TratarErrorDll (cod)
            FrmMain.Timer2.Enabled = False                                                 '// Se deshabilitan mensajes al usuario
            Atomic = False                                                                              '// Fin de las operaciones atómicas
            Exit Sub
        End If
    
        DoEvents
    Loop Until (cod = 0)

    FrmMain.Timer2.Enabled = False                                                          '// Se deshabilitan mensajes al usuario

    cod = UltimaRespuesta(nrohand, respfiscal)                                          '// Respuesta impresora fiscal

    If (cod = 0) Then
        respuesta = StrZToVB(respfiscal)                                                       '// Convertir a string Visual Basic
        AnalizarStatus respuesta
    Else
        '// La función de la DLL retornó Error
        '// -------------------------------------------
        MsgBox "ERROR:  " & Str(cod) & "Esperando Respuesta..!!" & vbCrLf & _
                     "Comando Enviado = " & CambiarSep(strcmd), vbCritical, "Error..."
        TratarErrorDll (cod)
    End If
 
    Atomic = False                                                                                       '// Fin de las operaciones atómicas

End Sub


'// ---------------------------------------------------------
'// Errores devueltos por las funciones de la DLL
'// Están definidos en winfis.bas 
'// ---------------------------------------------------------
Public Const ERROR = -1                       '// Error general DLL.
Public Const ERR_HANDLER = -2           '// Handler inválido.
Public Const ERR_ATOMIC = -3             '// Intento de enviar un comando cuando se está procesando el anterior.
Public Const ERR_TIMEOUT = -4           '// Error de comunicaciones.
Public Const ERR_ALREADYOPEN = -5  '// Puerto ya abierto.
Public Const ERR_NOMEM = -6             '// Memoria host insuficiente.
Public Const ERR_NOTOPENYET = -7    '// Intento de usar un puerto no abierto.
Public Const ERR_INVALIDPTR = -8      '// La dirección del buffer de respuesta es inválida.
Public Const ERR_STATPRN = -9           '// El comando no finalizó, sino que volvió una respuesta tipo STAT_PRN.
Public Const ERR_ABORT = -10            '// El proceso en curso fue abortado


'//=========================================================================================
'// Rutina que indica las acciones a seguir luego de un error en las funciones de la DLL
'//
'//              coderr = código de error reportado por la función de la DLL
'//=========================================================================================
Public Sub TratarErrorDll(coderr As Integer)
    
    '// Diseñar la rutina de tratamiento de error que corresponda
    '// Tener presente el estado del software
    '// ------------------------------------------------------------------------
    Select Case coderr
        Case ERR_HANDLER, ERR_INVALIDPTR, ERR_NOMEM, ERROR, ERR_NOTOPENYET, ERR_ABORT
        Case ERR_TIMEOUT
        Case ERR_ALREADYOPEN, ERR_ATOMIC
        Case ERR_STATPRN
    End Select

End Sub


'//=========================================================================================
'// Función que dado un string con caracteres separadores de campo, los cambia por otro
'// La idea es que el nuevo separador sea imprimible.
'// Esta función cambia el ASCII 28 por un ASCII 124
'//
'//                   cadena = string con los separadores originales
'//=========================================================================================
Public Function CambiarSep(cadena As String) As String
Dim s As String                                                                                 '// Un caracter del string
Dim Salida As String                                                                         '// String con los separadores ya alterados
Dim ind As Integer                                                                            '// Para recorrer el string

    Salida = ""
    
    For ind = 1 To Len(cadena)                                                            '// Se recorre el string buscando el ASCII 28
        s = Mid$(cadena, ind, 1)                                                            '// Se obtiene un caracter del string
        
        If (s = Chr$(28)) Then
            Salida = Salida & "|"                                                              '// Si se encuentra el ASCII 28 se cambia
        Else
            Salida = Salida & s                                                                 '// En otro caso dejar como está
        End If
    Next ind
    
    CambiarSep = Salida                                                                      '// Se retorna el string modificado
    
End Function



'//=========================================================================================
'// Rutina que analiza los status de las respuestas. Pensar al software como una máquina de estados para 
'// decidir si un bit encontrado en 1 debe interpretarse como error, o no.
'//
'// En esta rutina habría que agregar que si la respuesta corresponde al comando StatusRquest, y el modelo de '// impresora fiscal los reporta, habría que analizar también los campos de status auxiliar y status de documento
'//=========================================================================================
Private Sub AnalizarStatus(cadena As String)
Dim stprn As Long                                                                         '// Status de PRN
Dim stfis As Long                                                                           '// Status Fiscal
Dim mask As Long                                                                         '// Máscara para recorrer los bits del status
Dim ind As Integer                                                                         '// Para usar como índice

    infodfe = Split(cadena, Chr$(28))                                                '// Se separa la respuesta en campos

    stprn = Val("&H" + infodfe(0))
    mask = 1                                                                                   '// Bit cero, en 1

    For ind = 1 To 16                                                                      '// Se recorren los bits del Status PRN
        If ((stprn And mask) <> 0) Then                                             '// Está el bit en 1 ??
              '// Decidir si es un error o no, y como tratarlo
              '// Tener presente el estado del software
        End If
        
        mask = mask * 2                                                  '// Se desplaza el bit de la máscara un lugar a la izquierda
    Next

    stfis = Val("&H" + infodfe(1))
    mask = 1                                                                                 '// Bit cero, en 1

    For ind = 1 To 16                                                                     '// Se recorren los bits del Status Fiscal
        If ((stfis And mask) <> 0) Then                                             '// Está el bit en 1 ??
               '// Decidir si es un error o no, y como tratarlo
               '// Tener presente el estado del software
        End If
        
        mask = mask * 2                                              '// Se desplaza el bit de la máscara un lugar a la izquierda
    Next

End Sub


'//=========================================================================================
'// Función que dado un string finalizado en el caracter Null, lo convierte al formato de Visual Basic.
'//=========================================================================================
Private Function StrZToVB(cadena As String) As String
Dim a As Integer

    For a = 1 To Len(cadena)
        If Mid$(cadena, a, 1) = Chr$(0) Then Exit For                     '// Buscando el Null
    Next a

    StrZToVB = Left$(cadena, a - 1)                                              '// Se retornan los ( a-1 ) caracteres previos
                                                                                                  '//  al caracter nulo
End Function










En base a todo lo anterior podría hacerse una función de armado de string de comando, como la siguiente:


'//========================================================================================
'// Rutina de impresión de ítem. Se genera el comando:  PrintLineItem
'//
'//     			handler	= Para indicar puerto serie a utilizar
'//     			texto  	= Texto a imprimir
'//     			cant   	= Cantidad de unidades
'//     			punit  	= Precio unitario
'//     			alic   	= Alícuota IVA
'//     			oper   	= Suma / Resta
'//     			impint 	= Impuestos Internos
'//     			dsp    	= Display opcional
'//     			pbase  	= Monto base / final
'//
'// Se asume que cada argumento fue verificado sintácticamente antes de llamar a esta rutina
'//========================================================================================


Public Sub PrintLineItem(handler As Long, texto As String, cant As String, punit As String, alic As String, _
                                       oper As String, ImpInt As String, dsp As String, pbase As String)
Dim id As String                                                                                                     '// ID comando
Dim comando As String                                                                                          '// Comando PrintLineItem
Dim FS As String                                                                                                    '// Separador de campos

    FS = Chr$(28)
    id = Chr$(66)

    comando = id & FS & texto & FS & cant & FS & punit & FS & alic & FS & oper & FS & ImpInt & FS & dsp _
                     & FS & pbase
    EnviarComando Handler, comando

End Sub

